Smoothed Spectra by Lag Window—LWIN

The program LWIN (Smoothed Spectra by Parzen's **L**ag **Win**dow) is a subroutine subprogram that smooths the Fourier or power spectrum of a given time history by using Parzen's lag window.

LWIN（Smoothed Spectra by **L**ag **Win**dow）

【Purpose】

To compute the smoothed Fourier and power spectra of a given set of equally spaced data using Parzen's lag window with a specified bandwidth in the frequency domain.

【Usage】

（1） How to connect

```
CALL LWIN (N, X, ND1, DT, F, G, ND2, NFOLD, DF, BAND)
```

Argument	Type	Parameter in calling program	Return Parameter
N	I	Total number of real data X	Unchanged
X	R 1-D array (ND1)	Equal interval real data	Unchanged
ND1	I	Dimension size of X in calling program (ND1 .LE. 8192)	Unchanged
DT	R	Time interval (unit : sec)	Unchanged
F	R 1-D array (ND2)	No need to input here	Smoothed Fourier Spectrum
G	R 1-D array (ND2)	No need to input here	Smoothed Power Spectrum
ND2	I	Dimension size of F and G in calling program	Unchanged
NFOLD	I	No need to input here	Total number of Fourier & Power Spectra
DF	R	No need to input here	Frequency interval of Fourier & Power Spectra (Unit : Hz)
BAND	R	Band width (unit: Hz)	Unchanged

(2) Necessary subroutines and function subprograms

FAST

(3) Remarks

i) $ND2$ must be greater than or equal to $NT/2+1$, where NT is the smallest power of 2 greater than N, or N if N is a power of 2.

ii) The censored width in the time-lag domain must be $BAND/ DF \geq 560 /151$ (about 3.71) so that it does not exceed the bending point.

iii) If the argument $BAND$ = 0.0 then no smoothing is performed.

【Calculation Method】

This program uses the Fast Fourier Transform program **FAST**. First, let the data x_m ($m = 1, 2,\cdots, N$) in the time interval Δt be the real part of a complex number with zero imaginary part, and add subsequent complex zeros until the number of N_{total}, which is the number of a power of 2 closest to N. Next, Fourier transform these complex number data to obtain the complex Fourier coefficients C_k. Then perform the inverse Fourier transform to obtain the autocovariance coefficients R_j.

$$R_j = \sum_{k=0}^{N-1} \left| C_k \right|^2 \mathrm{e}^{\mathrm{i}(2\pi kj/N)}$$

Convert the bandwidth for smoothing in the frequency domain, b (Hz), into a censoring width, u (sec), using the following equation.

$$b = \frac{280}{151u}$$

Parzen's lag window $w(\tau)$ is obtained by the following equation.

$$w(\tau) = \begin{cases} 1-6\left(\frac{\tau}{u}\right)^2 + 6\left(\frac{|\tau|}{u}\right)^3 & |\tau| \leq u/2 \\ 2\left(1-\frac{|\tau|}{u}\right)^3 & u/2 \leq |\tau| \leq u \\ 0 & |\tau| > u \end{cases}$$

This is followed by multiplication in the time shift domain.

$$\overline{R}_j = R_j \cdot w(\tau)$$

If we Fourier transform $\overline{R_j}$ again, we get the smoothed Fourier spectrum, which in turn gives us the smoothed power spectrum.

When the bandwidth is specified as zero, the unsmoothed Fourier spectrum and power spectrum are obtained. This program has the same functions as the separate program **FPAC**, except that it computes the autocorrelation function.

【Program List】

```
C  * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * *
C     SUBROUTINE FOR SMOOTHED SPECTRA BY PARZEN'S LAG WINDOW
C  * * * * * * * * * * * * * * * * * * * * * * * * * * * * * * *
C
C                                          CODED BY Y.OHSAKI
C
C     PURPOSE
C       TO COMPUTE FOURIER AND POWER SPECTRA OF A SERIES OF EQUI-
C       SPACED DATA SMOOTHED BY APPLICATION OF PARZEN'S LAG WINDOW
C       WITH SPECIFIED BANDWIDTH IN FREQUENCY DOMAIN
C
C     USAGE
C       CALL LWIN(N,X,ND1,DT,F,G,ND2,NFOLD,DF,BAND)
C
C     DESCRIPTION OF ARGUMENTS
C       N       - TOTAL NUMBER OF DATA
C       X(ND1)  - EQUI-SPACED DATA
C       ND1     - DIMENSION OF X IN CALLING PROGRAM  ND1.LE.8192
C       DT      - TIME INCREMENT IN DATA IN SEC
C       F(ND2)  - SMOOTHED FOURIER SPECTRUM
C       G(ND2)  - SMOOTHED POWER SPECTRUM
C       ND2     - DIMENSION OF F,G IN CALLING PROGRAM
C       NFOLD   - NUMBER OF DATA IN THE SPECTRA
C       DF      - FREQUENCY INCREMENT IN THE SPECTRA IN HZ
C       BAND    - SPECIFIED BANDWIDTH IN HZ
C
C     REMARKS
C      (1) ND2.GE.NT/2+1, WHERE NT IS THE POWER OF 2 EQUAL TO N OR
C          MINIMUM LARGER THAN N
C      (2) BAND/DF MUST BE LARGER THAN 560/151 (APPROXIMATELY 3.71)
C      (3) IF THE ARGUMENT BAND=0., NO SMOOTHING IS MADE
C
C     SUBROUTINES AND FUNCTION SUBPROGRAMS REQUIRED
C       FAST
C
      SUBROUTINE LWIN(N,X,ND1,DT,F,G,ND2,NFOLD,DF,BAND)
C
      COMPLEX   A(8192)
      DIMENSION X(ND1),F(ND2),G(ND2)
C
C     INITIALIZATION
C
      DO 110 M=1,N
      A(M)=CMPLX(X(M),0.)
  110 CONTINUE
      NT=2
  120 IF(NT.GE.N) GO TO 130
      NT=NT*2
      GO TO 120
  130 IF(NT.EQ.N) GO TO 150
      DO 140 M=N+1,NT
      A(M)=(0.,0.)
```

```
  140 CONTINUE
  150 NFOLD=NT/2+1
      T=REAL(NT)*DT
      DF=1./T
C
C     AUTOCORRELATION
C
      CALL FAST(NT,A,8192,-1)
      IF(BAND.NE.0.) GO TO 170
      DO 160 K=1,NFOLD
      F(K)=CABS(A(K))*DT
  160 CONTINUE
      GO TO 240
  170 DO 180 K=1,NT
      A(K)=A(K)*CONJG(A(K))/REAL(NT)**2
  180 CONTINUE
      CALL FAST(NT,A,8192,+1)
C
C     LAG WINDOW
C
      U=3.708609/BAND*DF
      IF(U.GT.1.) GO TO 260
      U=REAL(NFOLD-1)*U
      DO 210 J=2,NFOLD
      TAU=REAL(J-1)/U
      IF(TAU.GT.0.5) GO TO 190
      A(J)=A(J)*(1.-6.*TAU**2*(1.-TAU))
      GO TO 210
  190 IF(TAU.GT.1.) GO TO 200
      A(J)=A(J)*2.*(1.-TAU)**3
      GO TO 210
  200 A(J)=(0.,0.)
  210 CONTINUE
      DO 220 J=2,NFOLD-1
      A(NT+2-J)=A(J)
  220 CONTINUE
C
C     SMOOTHED SPECTRA
C
      CALL FAST(NT,A,8192,-1)
      DO 230 K=1,NFOLD
      F(K)=SQRT(ABS(REAL(A(K)))*REAL(NT))*DT
  230 CONTINUE
  240 G(1)=F(1)**2/T
      DO 250 K=2,NFOLD-1
      G(K)=2.*F(K)**2/T
  250 CONTINUE
      G(NFOLD)=F(NFOLD)**2/T
      RETURN
C
  260 WRITE(6,601)
      STOP
C
```

```
C     FORMAT STATEMENT                                                  LWIN 106
C                                                                       LWIN 107
  601 FORMAT('BANDWIDTH IS TOO NARROW')                                 LWIN 108
      END                                                               LWIN 109
```

【Example】

From the file EQ.01, read the time interval, the number of seismic motion data, and the data and compute the unsmoothed power spectrum, and the smoothed power spectrum with a bandwidth of 0.8 Hz.

```
C
      DIMENSION DATA(800),F(513),G1(513),G2(513)
C
      READ(5,501) DT,NN,(DATA(M),M=1,NN)
      CALL LWIN(NN,DATA,800,DT,F,G1,513,NFOLD,DF,0.0)
      CALL LWIN(NN,DATA,800,DT,F,G2,513,NFOLD,DF,0.8)
C
      STOP
  501 FORMAT(T51,F10.0,I10/(8F10.0))
     END
```

Output：

The power spectrum is stored in array *G*1, and the smoothed power spectrum is stored in array *G*2, as shown in the following figure, which are represented by thin and thick lines, respectively.